
Section S1. Statistical Model8

We were interested in understanding the probability of occurrence of coral reef oases, and9

in particular, how environmental covariates mediated their probability of occurrence. As10

stated in the Methods, we defined an oasis as a reef site that exhibited higher coral cover11

relative to reef sites within a defined area. These areas were divided into 2.5 arcminute grid12

cells (~21.2 km2), and individual sites (< 100 m2) were sampled within these grid cells. The13

probability of occurrence can be separated into the true probability of occurrence of an oasis14

(ψ), and the probability of detecting an oasis (p). The true probability of occurrence (ψ)15

was modeled using a deterministic equation with the pertinent environmental covariates,16

and the detection probability (p) was modeled using a binomial distribution with the17

number of trials (sampled reefs) and successes (oases). Our modeling approach is based18

on species occupancy models (Mackenzie et al. 2002), which estimate species occupancy19

(i.e., occurrence) when detection probabilities are less than one.20

Our data set was the number of oases (yij) observed in a given cell i nested within sub-21

region j, given nij sampling occasions. We wished to predict the true probability of22

occurrence of an oasis for cell i. We defined the unobserved, true state of cell i as zij = 1 if it23

had an oasis, and zij = 0 if it did not. Then we modeled the data yij, the number of times24

we observed an oasis given nij sampling occasions as:25

yij ∼


0, if zij = 0

Binomial(nij, pj), if zij = 1

which states that we will never detect an oasis site if the cell does not have one, but if the26

cell does have an oasis, we will detect it with probability pj, estimated as a group-level27

intercept for sub-region j, designated as h(ηj) below.28

Next, we modeled the process governing the true state zij:29

2



zij ∼ Bernoulli(ψij)

We used a Bernoulli distribution because the random variable zij can take on values of 0 or30

1. The frequency of these values is determined by the true probability of occurrence, ψij.31

We modeled ψij using a deterministic model, g(αj, β, xij), where αj represented an intercept32

for sub-region j, β represented a vector of coefficients, and xij represented a vector of33

the measured covariates for cellij. The covariates were assumed to be measured without34

error and thus were not treated as random variables in our model. We used an inverse35

logit function because it returns continuous values from 0 to 1. Finally, we calculated the36

posterior probability of our random variables conditional on our data using the following37

Bayesian hierarchical model:38

[z, β, α, η, µα, σα, µη, ση|n, y] ∝
890

∏
i=1

32

∏
j=1

[yij|nij, h(ηij)zij] (detection model)

× [zij|g(αj, β, xij)] (occurrence model)

× [αj|µα, σα] (occurrence hyperprior)

× [ηj|µη, ση] (detection hyperprior)

× [β][µα][σα][µη][ση] (priors)

g(αj, β, xij) = invlogit(αj + βxij)

h(ηij) = invlogit(ηj)

with the following priors:39

β ∼ Normal(0, 1)

αj ∼ Normal(µα, σα)

µα ∼ Normal(−1, 1)

σα ∼ Exponential(1)

ηj ∼ Normal(µη, ση)

µη ∼ Normal(−1, 1)

ση ∼ Exponential(1)

We used weakly regularizing priors for the slope coefficients (β), noting that the envi-40

ronmental covariates were standardized to have a mean of 0 and standard deviation of 141

(Fig. S1A). We also chose weakly regularizing priors for the hyperpriors µα (Fig. S1B) and42

σα (Fig. S1C) so that their resulting group-level intercepts (αj) for the true probability of43

occurrence peaked between 0 and 0.3, and then declined steadily towards 1 (Fig. S2A). We44

chose to put more weight on probabilities less than 0.5 because oases are, by definition, rare45

occurrences. The same priors were used for hyperpriors µη and ση, for the same reasons.46

We visualized our prior predictive distributions to ensure that the resulting relationships47

between the true probability of occurrence and a standardized coefficient were reasonable48

(Fig. S2B).49

Section S2. Supporting Figures50

Figure S1. Predictive distributions based on 10000 simulated draws for the priors from the53

hierarchical model. Parameters are on the logit scale.54

Figure S2. Predictive distributions based on 10000 simulated draws for the priors from the56

hierarchical model. In (A), the probability of oasis occurrence is back-transformed from57

the logit scale. In (B), 25 randomly selected relationships between the true probability of58

occurrence and a standardized covariate derived from the prior predictive distributions.59

Section S3. JAGS model60

model{

# priors for occurrence model

for(i in 1:nX){

beta[i] ~ dnorm(0, 1)

}

# hyper-priors for occurrence model

mu.alpha ~ dnorm(-1, 1)

sigma.alpha ~ dexp(1)

tau.alpha <- 1/sigma.alpha^2

for(j in 1:y.n.sites){

alpha[j] ~ dnorm(mu.alpha, tau.alpha)

}

# hyper-priors for detection model

mu.det ~ dnorm(-1, 1)

sigma.det ~ dexp(1)

tau.det <- 1 / sigma.det^2

for(j in 1:y.n.sites){

eta[j] ~ dnorm(mu.det, sigma.det)

}

# likelihood

for(i in 1:N){

# occurrence model

logit(psi[i]) <- alpha[y.group[i]] + inprod(beta[], X[i, ] )

z[i] ~ dbern(psi[i])

# detection model

logit(p[i]) <- eta[y.group[i]]

mu.p[i] <- z[i] * p[i]

y[i] ~ dbin(mu.p[i], n[i])

# simulate new data, conditional on model parameters

y.new[i] ~ dbin(mu.p[i], n[i])

# pearson chi-square discrepancy for a binomial

# e is small value to avoid division by zero

chi2b.data[i] <- ((y[i] - mu.p[i] * n[i]) /

sqrt((mu.p[i] + e) * n[i] * (1 - mu.p[i] - e)))^2

chi2b.sim[i] <- ((y.new[i] - mu.p[i] * n[i]) /

sqrt((mu.p[i] + e) * n[i] * (1 - mu.p[i] - e)))^2

# freeman-tukey discrepancy for a binomial

ftd.data[i] <- (sqrt(y[i]) - sqrt(p[i] * z[i] * n[i]))^2

ftd.sim[i] <- (sqrt(y.new[i]) - sqrt(p[i] * z[i] * n[i]))^2

}

# bayesian p-value for chi-square discrepancy

d.chi2b.data <- sum(chi2b.data)

d.chi2b.sim <- sum(chi2b.sim)

p.chi2b <- step(d.chi2b.sim - d.chi2b.data)

# bayesian p-value for freeman-tukey discrepancy

d.ftd.data <- sum(ftd.data)

d.ftd.sim <- sum(ftd.sim)

p.ftd <- step(d.ftd.sim - d.ftd.data)

}

